# Rules for integrands of the form $(d\,x)^m\left(a+b\,x^n+c\,x^{2\,n}\right)^p$

x. $\int(d\,x)^m\left(b\,x^n+c\,x^{2\,n}\right)^p\,dx$

1. $\int(d\,x)^m\left(b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $p\in\mathbb{Z}$

1: $\int(d\,x)^m\left(b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $p\in\mathbb{Z}\land(m\in\mathbb{Z}\lor d>0)$

- Derivation: Algebraic simplification
- Basis: If $p\in\mathbb{Z}$, then $\left(b\,x^n+c\,x^{2\,n}\right)^p==x^{n\,p}\,(b+c\,x^n)^p$
- Rule 1.2.3.2.0.1.1: If $p\in\mathbb{Z}\land(m\in\mathbb{Z}\lor d>0)$, then

$$\int(d\,x)^m\left(b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow\;d^m\int x^{m+n\,p}\,(b+c\,x^n)^p\,dx$$

- Program code:

```
(* Int[(d_.*x_)^m_.*(b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  d^m*Int[x^(m+n*p)*(b+c*x^n)^p,x] /;
FreeQ[{b,c,d,m,n},x] && EqQ[n2,2*n] && IntegerQ[p] && (IntegerQ[m] || GtQ[d,0]) *)
```

2: $\int(d\,x)^m\left(b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $p\in\mathbb{Z}\land n\in\mathbb{Z}$

- Derivation: Algebraic simplification
- Basis: If $p\in\mathbb{Z}\land n\in\mathbb{Z}$, then $\left(b\,x^n+c\,x^{2\,n}\right)^p==\frac{1}{d^{n\,p}}\,(d\,x)^{n\,p}\,(b+c\,x^n)^p$
- Rule 1.2.3.2.0.1.2: If $p\in\mathbb{Z}\land n\in\mathbb{Z}$, then

$$\int(d\,x)^m\left(b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow\;\frac{1}{d^{n\,p}}\int(d\,x)^{m+n\,p}\,(b+c\,x^n)^p\,dx$$

- Program code:

```
(* Int[(d_.*x_)^m_.*(b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  1/d^(n*p)*Int[(d*x)^(m+n*p)*(b+c*x^n)^p,x] /;
FreeQ[{b,c,d,m},x] && EqQ[n2,2*n] && IntegerQ[p] && IntegerQ[n] *)
```

**3:** $\int (d\,x)^m \left(b\,x^n + c\,x^{2\,n}\right)^p\,dx$ when $p \in \mathbb{Z} \wedge \neg (m \in \mathbb{Z} \vee d > 0)$

**Derivation: Piecewise constant extraction**

**Basis:** $\partial_x \frac{(d\,x)^m}{x^m} == 0$

**Rule 1.2.3.2.0.1.3: If** $p \in \mathbb{Z} \wedge \neg (m \in \mathbb{Z} \vee d > 0)$, **then**

$$\int (d\,x)^m \left(b\,x^n + c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{(d\,x)^m}{x^m} \int x^{m+n\,p} \left(b + c\,x^n\right)^p\,dx$$

**Program code:**

```
(* Int[(d_.*x_)^m_.*(b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  (d*x)^m/x^m*Int[x^(m+n*p)*(b+c*x^n)^p,x] /;
FreeQ[{b,c,d,m,n},x] && EqQ[n2,2*n] && IntegerQ[p] && Not[IntegerQ[m] || GtQ[d,0]] *)
```

**2:** $\int (d\,x)^m \left(b\,x^n + c\,x^{2\,n}\right)^p\,dx$ when $p \notin \mathbb{Z}$

**Derivation: Piecewise constant extraction**

**Basis:** $\partial_x \frac{(b\,x^n+c\,x^{2\,n})^p}{(d\,x)^{n\,p}\,(b+c\,x^n)^p} == 0$

**Rule 1.2.3.2.0.2: If** $p \notin \mathbb{Z}$, **then**

$$\int (d\,x)^m \left(b\,x^n + c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{\left(b\,x^n + c\,x^{2\,n}\right)^p}{(d\,x)^{n\,p}\,\left(b + c\,x^n\right)^p} \int (d\,x)^{m+n\,p} \left(b + c\,x^n\right)^p\,dx$$

**Program code:**

```
(* Int[(d_.*x_)^m_.*(b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  (b*x^n+c*x^(2*n))^p/((d*x)^(n*p)*(b+c*x^n)^p)*Int[(d*x)^(m+n*p)*(b+c*x^2)^p,x] /;
FreeQ[{b,c,d,m,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[p]] *)
```

**1:** $\int x^m \left(a + b\,x^n + c\,x^{2\,n}\right)^p\,dx$ when $m - n + 1 == 0$

**Derivation: Integration by substitution**

**Basis:** $x^{n-1}\,F[x^n] == \frac{1}{n}\,\text{Subst}[F[x],\,x,\,x^n]\,\partial_x x^n$

**Rule 1.2.3.2.1: If** $m - n + 1 == 0$, **then**

$$\int x^m \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow \frac{1}{n} \text{Subst}\left[\int \left(a+b x+c x^2\right)^p dx,\ x,\ x^n\right]$$

**Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  1/n*Subst[Int[(a+b*x+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && EqQ[Simplify[m-n+1],0]
```

**2:** $\int (d x)^m \left(a+b x^n+c x^{2n}\right)^p dx$ when $p \in \mathbb{Z}^+$

**Derivation: Algebraic expansion**

**Rule 1.2.3.2.2: If $p \in \mathbb{Z}^+$, then**

$$\int (d x)^m \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow \int \text{ExpandIntegrand}\left[(d x)^m \left(a+b x^n+c x^{2n}\right)^p,\ x\right] dx$$

**Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d*x)^m*(a+b*x^n+c*x^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[n2,2*n] && IGtQ[p,0] && Not[IntegerQ[Simplify[(m+1)/n]]]
```

**3:** $\int x^m \left(a+b x^n+c x^{2n}\right)^p dx$ when $p \in \mathbb{Z}^- \land n < 0$

**Derivation: Algebraic simplification**

**Basis: If $p \in \mathbb{Z}$, then** $\left(a+b x^n+c x^{2n}\right)^p == x^{2np} \left(c+b x^{-n}+a x^{-2n}\right)^p$

**Rule 1.2.3.2.3: If $p \in \mathbb{Z}^- \land n < 0$, then**

$$\int x^m \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow \int x^{m+2np} \left(c+b x^{-n}+a x^{-2n}\right)^p dx$$

**Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  Int[x^(m+2*n*p)*(c+b*x^(-n)+a*x^(-2*n))^p,x] /;
FreeQ[{a,b,c,m,n},x] && EqQ[n2,2*n] && ILtQ[p,0] && NegQ[n]
```

4. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0$

x: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0\,\wedge\,p\in\mathbb{Z}$

**Derivation: Algebraic simplification**

**Basis: If $b^2-4\,a\,c==0$, then $a+b\,z+c\,z^2==\frac{1}{c}\left(\frac{b}{2}+c\,z\right)^2$**

**Rule 1.2.3.2.4.1: If $b^2-4\,a\,c==0\,\wedge\,p\in\mathbb{Z}$, then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{1}{c^p}\int (d\,x)^m \left(\frac{b}{2}+c\,x^n\right)^{2\,p}\,dx$$

**Program code:**

```
(* Int[(d_.*x_)^m_.*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/c^p*Int[(d*x)^m*(b/2+c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p] *)
```

2. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}$

x: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+2\,n\,(p+1)+1==0\,\wedge\,p\neq-\frac{1}{2}$

**Derivation: Square trinomial recurrence 2c with $m+2\,n\,(p+1)+1==0$**

**Rule 1.2.3.2.4.2.1: If $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m+2\,n\,(p+1)+1==0\,\wedge\,p\neq-\frac{1}{2}$, then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{(d\,x)^{m+1}\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{2\,a\,d\,n\,(p+1)\,(2\,p+1)}-\frac{(d\,x)^{m+1}\,(2\,a+b\,x^n)\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{2\,a\,d\,n\,(2\,p+1)}$$

**Program code:**

```
(* Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(2*a*d*n*(p+1)*(2*p+1)) -
  (d*x)^(m+1)*(2*a+b*x^n)*(a+b*x^n+c*x^(2*n))^p/(2*a*d*n*(2*p+1)) /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && EqQ[m+2*n*(p+1)+1,0] && NeQ[2*p+1,0] *)
```

2: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0\,\land\,p\notin\mathbb{Z}$

**Derivation: Piecewise constant extraction**

- **Basis:** If $b^2-4\,a\,c==0$, then $\partial_x \frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(1+\frac{2\,c\,x^n}{b}\right)^{2\,p}}==0$

- **Rule 1.2.3.2.4.2.2:** If $b^2-4\,a\,c==0\,\land\,p\notin\mathbb{Z}$, then

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{a^{\text{IntPart}[p]}\left(a+b\,x^n+c\,x^{2\,n}\right)^{\text{FracPart}[p]}}{\left(1+\frac{2\,c\,x^n}{b}\right)^{2\,\text{FracPart}[p]}}\int (d\,x)^m\left(1+\frac{2\,c\,x^n}{b}\right)^{2\,p}dx$$

**Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^FracPart[p]/(c^IntPart[p]*(b/2+c*x^n)^(2*FracPart[p]))*Int[(d*x)^m*(b/2+c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p-1/2]
```

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^n+c*x^(2*n))^FracPart[p]/(1+2*c*x^n/b)^(2*FracPart[p])*Int[(d*x)^m*(1+2*c*x^n/b)^(2*p),x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[2*p]]
```

5. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \bigwedge\ \frac{m+1}{n}\in\mathbb{Z}$

1: $\int x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \bigwedge\ \frac{m+1}{n}\in\mathbb{Z}$

**Derivation: Integration by substitution**

- **Basis: If** $\frac{m+1}{n}\in\mathbb{Z}$, **then** $x^m\,F[x^n] == \frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\ x,\ x^n\right]\,\partial_x x^n$
- **Note: If** $n\in\mathbb{Z}\ \bigwedge\ \frac{m+1}{n}\in\mathbb{Z}$, **then** $m\in\mathbb{Z}$, **and** $(d\,x)^m$ **automatically evaluates to** $d^m\,x^m$.
- **Rule 1.2.3.2.5.1: If** $b^2-4\,a\,c\neq 0\ \bigwedge\ \frac{m+1}{n}\in\mathbb{Z}$, **then**

$$\int x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\ \longrightarrow\ \frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\left(a+b\,x+c\,x^2\right)^p\,dx,\ x,\ x^n\right]$$

**Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*x+c*x^2)^p,x],x,x^n] /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[(m+1)/n]]
```

2: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \bigwedge\ \frac{m+1}{n}\in\mathbb{Z}$

**Derivation: Piecewise constant extraction**

- **Basis:** $\partial_x \frac{(d\,x)^m}{x^m} == 0$
- **Basis:** $\frac{(d\,x)^m}{x^m} == \frac{d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$
- **Rule 1.2.3.2.5.2: If** $b^2-4\,a\,c\neq 0\ \bigwedge\ \frac{m+1}{n}\in\mathbb{Z}$, **then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\ \longrightarrow\ \frac{d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

**Program code:**

```
Int[(d_*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  d^IntPart[m]*(d*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[(m+1)/n]]
```

6. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}$

1. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+$

1: $\int x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\wedge\ m\in\mathbb{Z}\ \wedge\ \text{GCD}[m+1,\ n]\neq 1$

**Derivation: Integration by substitution**

- **Basis: If** $n\in\mathbb{Z}\ \wedge\ m\in\mathbb{Z}$, **let** $k=\text{GCD}[m+1,\ n]$, **then** $x^m\,F[x^n]==\frac{1}{k}\,\text{Subst}\left[x^{\frac{m+1}{k}-1}\,F\left[x^{n/k}\right],\ x,\ x^k\right]\,\partial_x x^k$

**Rule 1.2.3.2.6.1.1: If** $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\wedge\ m\in\mathbb{Z}$, **let** $k=\text{GCD}[m+1,\ n]$, **if** $k\neq 1$, **then**

$$\int x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\ \longrightarrow\ \frac{1}{k}\,\text{Subst}\left[\int x^{\frac{m+1}{k}-1}\left(a+b\,x^{n/k}+c\,x^{2\,n/k}\right)^p\,dx,\ x,\ x^k\right]$$

**Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{k=GCD[m+1,n]},
  1/k*Subst[Int[x^((m+1)/k-1)*(a+b*x^(n/k)+c*x^(2*n/k))^p,x],x,x^k] /;
 k≠1] /;
FreeQ[{a,b,c,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IntegerQ[m]
```

2: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\wedge\ m\in\mathbb{F}$

**Derivation: Integration by substitution**

- **Basis: If** $k\in\mathbb{Z}^+$, **then** $(d\,x)^m\,F[x]==\frac{k}{d}\,\text{Subst}\left[x^{k\,(m+1)-1}\,F\left[\frac{x^k}{d}\right],\ x,\ (d\,x)^{1/k}\right]\,\partial_x (d\,x)^{1/k}$

**Rule 1.2.3.2.6.1.2: If** $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\wedge\ m\in\mathbb{F}$, **let** $k=\text{Denominator}[m]$, **then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\ \longrightarrow\ \frac{k}{d}\,\text{Subst}\left[\int x^{k\,(m+1)-1}\left(a+\frac{b\,x^{k\,n}}{d^n}+\frac{c\,x^{2\,k\,n}}{d^{2\,n}}\right)^p\,dx,\ x,\ (d\,x)^{1/k}\right]$$

**Program code:**

```
Int[(d_.*x_)^m_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  k/d*Subst[Int[x^(k*(m+1)-1)*(a+b*x^(k*n)/d^n+c*x^(2*k*n)/d^(2*n))^p,x],x,(d*x)^(1/k)]] /;
FreeQ[{a,b,c,d,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && FractionQ[m] && IntegerQ[p]
```

3. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;n\in\mathbb{Z}^+\;\wedge\;p\in\mathbb{Z}^+$

1: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;n\in\mathbb{Z}^+\;\wedge\;p\in\mathbb{Z}^+\;\wedge\;m>n-1\;\wedge\;m+2\,n\,p+1\neq 0\;\wedge\;m+n\,(2\,p-1)+1\neq 0$

**Derivation: Trinomial recurrence 1b with $A=0$, $B=1$ and $m=m-n$**

**Rule 1.2.3.2.6.1.3.1: If $b^2-4\,a\,c\neq 0\;\wedge\;n\in\mathbb{Z}^+\;\wedge\;p\in\mathbb{Z}^+\;\wedge\;m>n-1\;\wedge\;m+2\,n\,p+1\neq 0\;\wedge\;m+n\,(2\,p-1)+1\neq 0$, then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow$$

$$\frac{d^{n-1}\,(d\,x)^{m-n+1}\left(a+b\,x^n+c\,x^{2\,n}\right)^p\left(b\,n\,p+c\,(m+n\,(2\,p-1)+1)\,x^n\right)}{c\,(m+2\,n\,p+1)\,(m+n\,(2\,p-1)+1)}\,-$$

$$\frac{n\,p\,d^n}{c\,(m+2\,n\,p+1)\,(m+n\,(2\,p-1)+1)}\int (d\,x)^{m-n}\left(a+b\,x^n+c\,x^{2\,n}\right)^{p-1}\left(a\,b\,(m-n+1)-\left(2\,a\,c\,(m+n\,(2\,p-1)+1)-b^2\,(m+n\,(p-1)+1)\right)x^n\right)dx$$

**Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  d^(n-1)*(d*x)^(m-n+1)*(a+b*x^n+c*x^(2*n))^p*(b*n*p+c*(m+n*(2*p-1)+1)*x^n)/(c*(m+2*n*p+1)*(m+n*(2*p-1)+1)) -
  n*p*d^n/(c*(m+2*n*p+1)*(m+n*(2*p-1)+1))*
    Int[(d*x)^(m-n)*(a+b*x^n+c*x^(2*n))^(p-1)*Simp[a*b*(m-n+1)-(2*a*c*(m+n*(2*p-1)+1)-b^2*(m+n*(p-1)+1))*x^n,x],x] /;
FreeQ[{a,b,c,d},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IGtQ[p,0] && GtQ[m,n-1] && NeQ[m+2*n*p+1,0] && NeQ[m+n*(2*p-1
```

2: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\;\wedge\;n\in\mathbb{Z}^+\;\wedge\;p\in\mathbb{Z}^+\;\wedge\;m<-1$

**Reference: G&R 2.160.2**

**Derivation: Trinomial recurrence 1a with $A=1$ and $B=0$**

**Rule 1.2.3.2.6.1.3.2: If $b^2-4\,a\,c\neq 0\;\wedge\;n\in\mathbb{Z}^+\;\wedge\;p\in\mathbb{Z}^+\;\wedge\;m<-1$, then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow\;\frac{(d\,x)^{m+1}\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{d\,(m+1)}-\frac{n\,p}{d^n\,(m+1)}\int (d\,x)^{m+n}\,(b+2\,c\,x^n)\left(a+b\,x^n+c\,x^{2\,n}\right)^{p-1}dx$$

**Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^n+c*x^(2*n))^p/(d*(m+1)) -
  n*p/(d^n*(m+1))*Int[(d*x)^(m+n)*(b+2*c*x^n)*(a+b*x^n+c*x^(2*n))^(p-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IGtQ[p,0] && LtQ[m,-1]
```

3: $\int (d\,x)^m \left(a + b\,x^n + c\,x^{2\,n}\right)^p \,dx$ when $b^2 - 4\,a\,c \neq 0 \land n \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+ \land m + 2\,n\,p + 1 \neq 0$

**Derivation: Trinomial recurrence 1a with $A = 0$, $B = 1$ and $m = m - n$**

**Derivation: Trinomial recurrence 1b with $A = 1$ and $B = 0$**

**Rule 1.2.3.2.6.1.3.4: If $b^2 - 4\,a\,c \neq 0 \land n \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+ \land m + 2\,n\,p + 1 \neq 0$, then**

$$\int (d\,x)^m \left(a + b\,x^n + c\,x^{2\,n}\right)^p \,dx \longrightarrow \frac{(d\,x)^{m+1} \left(a + b\,x^n + c\,x^{2\,n}\right)^p}{d\,(m + 2\,n\,p + 1)} + \frac{n\,p}{m + 2\,n\,p + 1} \int (d\,x)^m \,(2\,a + b\,x^n) \left(a + b\,x^n + c\,x^{2\,n}\right)^{p-1} dx$$

**Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^n+c*x^(2*n))^p/(d*(m+2*n*p+1)) +
  n*p/(m+2*n*p+1)*Int[(d*x)^m*(2*a+b*x^n)*(a+b*x^n+c*x^(2*n))^(p-1),x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IGtQ[p,0] && NeQ[m+2*n*p+1,0]
```

4. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p \, dx$ when $b^2-4\,a\,c\neq 0\ \land\ n\in\mathbb{Z}^+\ \land\ p+1\in\mathbb{Z}^-$

1. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p \, dx$ when $b^2-4\,a\,c\neq 0\ \land\ n\in\mathbb{Z}^+\ \land\ p+1\in\mathbb{Z}^-\ \land\ m>n-1$

1: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p \, dx$ when $b^2-4\,a\,c\neq 0\ \land\ n\in\mathbb{Z}^+\ \land\ p+1\in\mathbb{Z}^-\ \land\ n-1<m\leq 2\,n-1$

**Derivation: Trinomial recurrence 2a with $A=1$ and $B=0$**

**Derivation: Trinomial recurrence 2b with $A=0$, $B=1$ and $m=m-n$**

**Rule 1.2.3.2.6.1.4.1.1: If $b^2-4\,a\,c\neq 0\ \land\ n\in\mathbb{Z}^+\ \land\ p+1\in\mathbb{Z}^-\ \land\ n-1<m\leq 2\,n-1$, then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p \, dx \longrightarrow$$

$$\frac{d^{n-1}\,(d\,x)^{m-n+1}\,(b+2\,c\,x^n)\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{n\,(p+1)\left(b^2-4\,a\,c\right)}-$$

$$\frac{d^n}{n\,(p+1)\left(b^2-4\,a\,c\right)}\int (d\,x)^{m-n}\,\left(b\,(m-n+1)+2\,c\,(m+2\,n\,(p+1)+1)\,x^n\right)\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1} dx$$

**Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  d^(n-1)*(d*x)^(m-n+1)*(b+2*c*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(n*(p+1)*(b^2-4*a*c)) -
  d^n/(n*(p+1)*(b^2-4*a*c))*
    Int[(d*x)^(m-n)*(b*(m-n+1)+2*c*(m+2*n*(p+1)+1)*x^n)*(a+b*x^n+c*x^(2*n))^(p+1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && ILtQ[p,-1] && GtQ[m,n-1] && LeQ[m,2*n-1]
```

2: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p \, dx$ when $b^2-4\,a\,c\neq 0\ \land\ n\in\mathbb{Z}^+\ \land\ p+1\in\mathbb{Z}^-\ \land\ m>2\,n-1$

**Derivation: Trinomial recurrence 2a with $A=0$, $B=1$ and $m=m-n$**

**Rule 1.2.3.2.6.1.4.1.2: If $b^2-4\,a\,c\neq 0\ \land\ n\in\mathbb{Z}^+\ \land\ p+1\in\mathbb{Z}^-\ \land\ m>2\,n-1$, then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p \, dx \longrightarrow$$

$$-\frac{d^{2\,n-1}\,(d\,x)^{m-2\,n+1}\,(2\,a+b\,x^n)\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{n\,(p+1)\left(b^2-4\,a\,c\right)}+$$

$$\frac{d^{2n}}{n(p+1)(b^2-4ac)}\int (d x)^{m-2n}\left(2a(m-2n+1)+b(m+n(2p+1)+1)x^n\right)\left(a+b x^n+c x^{2n}\right)^{p+1}dx$$

– **Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  -d^(2*n-1)*(d*x)^(m-2*n+1)*(2*a+b*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(n*(p+1)*(b^2-4*a*c)) +
  d^(2*n)/(n*(p+1)*(b^2-4*a*c))*
    Int[(d*x)^(m-2*n)*(2*a*(m-2*n+1)+b*(m+n*(2*p+1)+1)*x^n)*(a+b*x^n+c*x^(2*n))^(p+1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && ILtQ[p,-1] && GtQ[m,2*n-1]
```

2: $\int (d x)^m \left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4ac\neq 0 \land n\in\mathbb{Z}^+ \land p+1\in\mathbb{Z}^-$

**Derivation: Trinomial recurrence 2b with $A = 1$ and $B = 0$**

– **Rule 1.2.3.2.6.1.4.2: If $b^2-4ac\neq 0 \land n\in\mathbb{Z}^+ \land p+1\in\mathbb{Z}^-$, then**

$$\int (d x)^m \left(a+b x^n+c x^{2n}\right)^p dx \longrightarrow$$

$$-\frac{(d x)^{m+1}\left(b^2-2ac+b c x^n\right)\left(a+b x^n+c x^{2n}\right)^{p+1}}{a d n (p+1)\left(b^2-4ac\right)}+$$

$$\frac{1}{a n (p+1)\left(b^2-4ac\right)}\int (d x)^m \left(a+b x^n+c x^{2n}\right)^{p+1}\left(b^2(m+n(p+1)+1)-2ac(m+2n(p+1)+1)+b c(m+n(2p+3)+1)x^n\right)dx$$

– **Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  -(d*x)^(m+1)*(b^2-2*a*c+b*c*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*d*n*(p+1)*(b^2-4*a*c)) +
  1/(a*n*(p+1)*(b^2-4*a*c))*
    Int[(d*x)^m*(a+b*x^n+c*x^(2*n))^(p+1)*Simp[b^2*(m+n*(p+1)+1)-2*a*c*(m+2*n*(p+1)+1)+b*c*(m+n*(2*p+3)+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && ILtQ[p,-1]
```

5: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\bigwedge\,n\in\mathbb{Z}^+\,\bigwedge\,m>2\,n-1\,\bigwedge\,m+2\,n\,p+1\neq 0$

**Reference: G&R 2.160.3**

**Derivation: Trinomial recurrence 3a with $A=0$, $B=1$ and $m=m-n$**

**Note: G&R 2.174.1 is a special case of G&R 2.160.3.**

**Rule 1.2.3.2.6.1.5: If $b^2-4\,a\,c\neq 0\,\bigwedge\,n\in\mathbb{Z}^+\,\bigwedge\,m>2\,n-1\,\bigwedge\,m+2\,n\,p+1\neq 0$, then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow$$

$$\frac{d^{2\,n-1}\,(d\,x)^{m-2\,n+1}\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{c\,(m+2\,n\,p+1)}-\frac{d^{2\,n}}{c\,(m+2\,n\,p+1)}\int (d\,x)^{m-2\,n}\,(a\,(m-2\,n+1)+b\,(m+n\,(p-1)+1)\,x^n)\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

**Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  d^(2*n-1)*(d*x)^(m-2*n+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(c*(m+2*n*p+1)) -
  d^(2*n)/(c*(m+2*n*p+1))*
    Int[(d*x)^(m-2*n)*Simp[a*(m-2*n+1)+b*(m+n*(p-1)+1)*x^n,x]*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[m,2*n-1] && NeQ[m+2*n*p+1,0] && IntegerQ[p]
```

6: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ m<-1$

**Reference: G&R 2.160.1**

**Derivation: Trinomial recurrence 3b with $A=1$ and $B=0$**

**Note: G&R 2.161.6 is a special case of G&R 2.160.1.**

**Rule 1.2.3.2.6.1.6: If $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ m<-1$, then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\ \longrightarrow$$

$$\frac{(d\,x)^{m+1}\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{a\,d\,(m+1)}-\frac{1}{a\,d^n\,(m+1)}\int (d\,x)^{m+n}\left(b\,(m+n\,(p+1)+1)+c\,(m+2\,n\,(p+1)+1)\,x^n\right)\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

**Program code:**

```
Int[(d_.*x_)^m_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*d*(m+1)) -
  1/(a*d^n*(m+1))*Int[(d*x)^(m+n)*(b*(m+n*(p+1)+1)+c*(m+2*n*(p+1)+1)*x^n)*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[m,-1] && IntegerQ[p]
```

7. $\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+$

1: $\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ m<-1$

**Reference: G&R 2.176, CRC 123**

**Derivation: Algebraic expansion**

**Basis:** $\frac{(d\,z)^m}{a+b\,z+c\,z^2}==\frac{(d\,z)^m}{a}-\frac{1}{a\,d}\,\frac{(d\,z)^{m+1}\,(b+c\,z)}{a+b\,z+c\,z^2}$

**Rule 1.2.3.2.6.1.7.1: If** $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ m<-1$**, then**

$$\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx\ \longrightarrow\ \frac{(d\,x)^{m+1}}{a\,d\,(m+1)}-\frac{1}{a\,d^n}\int \frac{(d\,x)^{m+n}\,(b+c\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx$$

**Program code:**

```
Int[(d_.*x_)^m_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  (d*x)^(m+1)/(a*d*(m+1)) -
  1/(a*d^n)*Int[(d*x)^(m+n)*(b+c*x^n)/(a+b*x^n+c*x^(2*n)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[m,-1]
```

2. $\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ m>2\,n-1$

1: $\int \frac{x^m}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ m>3\,n-1\ \wedge\ m\in\mathbb{Z}$

**Derivation: Algebraic expansion**

**Rule 1.2.3.2.6.1.7.2.1: If** $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ m>3\,n-1\ \wedge\ m\in\mathbb{Z}$**, then**

$$\int \frac{x^m}{a+b\,x^n+c\,x^{2\,n}}\,dx\ \longrightarrow\ \int \text{PolynomialDivide}\left[x^m,\ a+b\,x^n+c\,x^{2\,n},\ x\right]\,dx$$

**Program code:**

```
Int[x_^m_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  Int[PolynomialDivide[x^m,(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && IGtQ[m,3*n-1]
```

2: $\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ m>2\,n-1$ **Not necessary?**

- **Reference: G&R 2.174.1, CRC 119**

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{(d\,z)^m}{a+b\,z+c\,z^2}==\frac{d^2\,(d\,z)^{m-2}}{c}-\frac{d^2}{c}\,\frac{(d\,z)^{m-2}\,(a+b\,z)}{a+b\,z+c\,z^2}$

- **Rule 1.2.3.2.6.1.7.2.2: If** $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ m>2\,n-1$**, then**

$$\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx \longrightarrow \frac{d^{2\,n-1}\,(d\,x)^{m-2\,n+1}}{c\,(m-2\,n+1)}-\frac{d^{2\,n}}{c}\int \frac{(d\,x)^{m-2\,n}\,(a+b\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx$$

**Program code:**

```
Int[(d_.*x_)^m_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  d^(2*n-1)*(d*x)^(m-2*n+1)/(c*(m-2*n+1)) -
  d^(2*n)/c*Int[(d*x)^(m-2*n)*(a+b*x^n)/(a+b*x^n+c*x^(2*n)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[m,2*n-1]
```

3. $\int \frac{x^m}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac\neq 0 \bigwedge \left(\frac{n}{2} \mid m\right) \in \mathbb{Z}^+ \bigwedge \frac{n}{2}\leq m<2n \bigwedge b^2-4ac \not> 0$

1: $\int \frac{x^m}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac\neq 0 \bigwedge \left(\frac{n}{2} \mid m\right) \in \mathbb{Z}^+ \bigwedge \frac{3n}{2}\leq m<2n \bigwedge b^2-4ac \not> 0$

**Derivation: Algebraic expansion**

- **Basis: If** $q\to\sqrt{\frac{a}{c}}$ **and** $r\to\sqrt{2q-\frac{b}{c}}$**, then** $\frac{z^3}{a+b z^2+c z^4}==\frac{q+r z}{2 c r\,(q+r z+z^2)}-\frac{q-r z}{2 c r\,(q-r z+z^2)}$

- **Note: If** $(a \mid b \mid c)\in\mathbb{R} \bigwedge b^2-4ac<0$**, then** $\frac{a}{c}>0$ **and** $2\sqrt{\frac{a}{c}}-\frac{b}{c}>0$.

- **Rule 1.2.3.2.6.1.7.3.1: If** $b^2-4ac\neq 0 \bigwedge \left(\frac{n}{2} \mid m\right) \in \mathbb{Z}^+ \bigwedge \frac{3n}{2}\leq m<2n \bigwedge b^2-4ac \not> 0$**, let** $q\to\sqrt{\frac{a}{c}}$ **and** $r\to\sqrt{2q-\frac{b}{c}}$**, then**

$$\int \frac{x^m}{a+b x^n+c x^{2n}} \, dx \longrightarrow \frac{1}{2 c r}\int \frac{x^{m-3n/2}\left(q+r x^{n/2}\right)}{q+r x^{n/2}+x^n}\, dx-\frac{1}{2 c r}\int \frac{x^{m-3n/2}\left(q-r x^{n/2}\right)}{q-r x^{n/2}+x^n}\, dx$$

**Program code:**

```
Int[x_^m_./(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  With[{q=Rt[a/c,2]},
  With[{r=Rt[2*q-b/c,2]},
  1/(2*c*r)*Int[x^(m-3*(n/2))*(q+r*x^(n/2))/(q+r*x^(n/2)+x^n),x] -
  1/(2*c*r)*Int[x^(m-3*(n/2))*(q-r*x^(n/2))/(q-r*x^(n/2)+x^n),x]]] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n/2,0] && IGtQ[m,0] && GeQ[m,3*n/2] && LtQ[m,2*n] && NegQ[b^2-4*a*c]
```

2: $\int \frac{x^m}{a+b x^n+c x^{2n}} \, dx$ when $b^2-4ac \neq 0 \bigwedge \left(\frac{n}{2} \mid m\right) \in \mathbb{Z}^+ \bigwedge \frac{n}{2} \leq m < \frac{3n}{2} \bigwedge b^2-4ac \not> 0$

**Derivation: Algebraic expansion**

- **Basis: If** $q \to \sqrt{\frac{a}{c}}$ **and** $r \to \sqrt{2q-\frac{b}{c}}$, **then** $\frac{z}{a+b z^2+c z^4} == \frac{1}{2cr\,(q-r z+z^2)} - \frac{1}{2cr\,(q+r z+z^2)}$
- **Note: If** $(a \mid b \mid c) \in \mathbb{R} \bigwedge b^2-4ac<0$, **then** $\frac{a}{c}>0$ **and** $2\sqrt{\frac{a}{c}}-\frac{b}{c}>0$.
- **Rule 1.2.3.2.6.1.7.3.2: If** $b^2-4ac \neq 0 \bigwedge \left(\frac{n}{2} \mid m\right) \in \mathbb{Z}^+ \bigwedge \frac{n}{2} \leq m < \frac{3n}{2} \bigwedge b^2-4ac \not> 0$, **let** $q \to \sqrt{\frac{a}{c}}$ **and** $r \to \sqrt{2q-\frac{b}{c}}$, **then**

$$\int \frac{x^m}{a+b x^n+c x^{2n}} \, dx \longrightarrow \frac{1}{2cr} \int \frac{x^{m-n/2}}{q-r x^{n/2}+x^n} \, dx - \frac{1}{2cr} \int \frac{x^{m-n/2}}{q+r x^{n/2}+x^n} \, dx$$

**Program code:**

```
Int[x_^m_./(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  With[{q=Rt[a/c,2]},
  With[{r=Rt[2*q-b/c,2]},
  1/(2*c*r)*Int[x^(m-n/2)/(q-r*x^(n/2)+x^n),x] -
  1/(2*c*r)*Int[x^(m-n/2)/(q+r*x^(n/2)+x^n),x]]] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n/2,0] && IGtQ[m,0] && GeQ[m,n/2] && LtQ[m,3*n/2] && NegQ[b^2-4*a*c]
```

4: $\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,m\geq n$

**Reference: G&R 2.161.1a & G&R 2.161.3**

**Derivation: Algebraic expansion**

- **Basis: Let** $q\to\sqrt{b^2-4\,a\,c}$**, then** $\frac{(d\,z)^m}{a+b\,z+c\,z^2}==\frac{d}{2}\left(\frac{b}{q}+1\right)\frac{(d\,z)^{m-1}}{\frac{b}{2}+\frac{q}{2}+c\,z}-\frac{d}{2}\left(\frac{b}{q}-1\right)\frac{(d\,z)^{m-1}}{\frac{b}{2}-\frac{q}{2}+c\,z}$

- **Rule 1.2.3.2.6.1.7.4: If** $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^+\,\wedge\,m\geq n$**, let** $q\to\sqrt{b^2-4\,a\,c}$**, then**

$$\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,\frac{d^n}{2}\left(\frac{b}{q}+1\right)\int\frac{(d\,x)^{m-n}}{\frac{b}{2}+\frac{q}{2}+c\,x^n}\,dx-\frac{d^n}{2}\left(\frac{b}{q}-1\right)\int\frac{(d\,x)^{m-n}}{\frac{b}{2}-\frac{q}{2}+c\,x^n}\,dx$$

**Program code:**

```
Int[(d_.*x_)^m_/(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  d^n/2*(b/q+1)*Int[(d*x)^(m-n)/(b/2+q/2+c*x^n),x] -
  d^n/2*(b/q-1)*Int[(d*x)^(m-n)/(b/2-q/2+c*x^n),x]] /;
FreeQ[{a,b,c,d},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GeQ[m,n]
```

5: $\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+$

**Reference: G&R 2.161.1a**

**Derivation: Algebraic expansion**

- **Basis: Let** $q\to\sqrt{b^2-4\,a\,c}$ **, then** $\frac{1}{a+b\,z+c\,z^2}==\frac{c}{q}\,\frac{1}{\frac{b}{2}-\frac{q}{2}+c\,z}-\frac{c}{q}\,\frac{1}{\frac{b}{2}+\frac{q}{2}+c\,z}$

- **Rule 1.2.3.2.6.1.7.5: If** $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+$**, let** $q\to\sqrt{b^2-4\,a\,c}$ **, then**

$$\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx\;\longrightarrow\;\frac{c}{q}\int\frac{(d\,x)^m}{\frac{b}{2}-\frac{q}{2}+c\,x^n}\,dx-\frac{c}{q}\int\frac{(d\,x)^m}{\frac{b}{2}+\frac{q}{2}+c\,x^n}\,dx$$

**Program code:**

```
Int[(d_.*x_)^m_./(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  c/q*Int[(d*x)^m/(b/2-q/2+c*x^n),x] - c/q*Int[(d*x)^m/(b/2+q/2+c*x^n),x]] /;
FreeQ[{a,b,c,d,m},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0]
```

2. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-$

1. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\wedge\,m\in\mathbb{Q}$

1: $\int x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\wedge\,m\in\mathbb{Z}$

**Derivation: Integration by substitution**

**Basis:** $F[x] = -\text{Subst}\left[\frac{F[x^{-1}]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

**Rule 1.2.3.2.6.2.1.1: If** $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\wedge\,m\in\mathbb{Z}$**, then**

$$\int x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \to -\text{Subst}\left[\int\frac{\left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p}{x^{m+2}}dx,\,x,\,\frac{1}{x}\right]$$

**Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  -Subst[Int[(a+b*x^(-n)+c*x^(-2*n))^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && ILtQ[n,0] && IntegerQ[m]
```

2: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\wedge\,m\in\mathbb{F}$

**Derivation: Integration by substitution**

**Basis: If** $n\in\mathbb{Z}\,\wedge\,k>1$**, then** $(d\,x)^m F[x^n] = -\frac{k}{d}\,\text{Subst}\left[\frac{F[d^{-n}\,x^{-k\,n}]}{x^{k\,(m+1)+1}},\,x,\,\frac{1}{(d\,x)^{1/k}}\right]\partial_x\frac{1}{(d\,x)^{1/k}}$

**Rule 1.2.3.2.6.2.1.2: If** $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{Z}^-\wedge\,m\in\mathbb{F}$**, let** $k=\text{Denominator}[m]$**, then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx \to -\frac{k}{d}\,\text{Subst}\left[\int\frac{\left(a+b\,d^{-n}\,x^{-k\,n}+c\,d^{-2\,n}\,x^{-2\,k\,n}\right)^p}{x^{k\,(m+1)+1}}dx,\,x,\,\frac{1}{(d\,x)^{1/k}}\right]$$

**Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  -k/d*Subst[Int[(a+b*d^(-n)*x^(-k*n)+c*d^(-2*n)*x^(-2*k*n))^p/x^(k*(m+1)+1),x],x,1/(d*x)^(1/k)]] /;
FreeQ[{a,b,c,d,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && ILtQ[n,0] && FractionQ[m]
```

**2:** $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^-\ \wedge\ m\notin\mathbb{Q}$

**Derivation: Piecewise constant extraction and integration by substitution**

- **Basis:** $\partial_x\left((d\,x)^m\left(x^{-1}\right)^m\right)=0$
- **Basis:** $(d\,x)^m\left(x^{-1}\right)^m=d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}\left(x^{-1}\right)^{\text{FracPart}[m]}$
- **Basis:** $F[x]=-\text{Subst}\left[\frac{F[x^{-1}]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$
- **Rule 1.2.3.2.6.2.2: If** $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^-\ \wedge\ m\notin\mathbb{Q}$**, then**

$$\int (d\,x)^m\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\ \longrightarrow\ d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}\left(x^{-1}\right)^{\text{FracPart}[m]}\int\frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(x^{-1}\right)^m}\,dx$$

$$\longrightarrow\ -d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}\left(x^{-1}\right)^{\text{FracPart}[m]}\,\text{Subst}\left[\int\frac{\left(a+b\,x^{-n}+c\,x^{-2\,n}\right)^p}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

**Program code:**

```
Int[(d_.*x_)^m_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  -d^IntPart[m]*(d*x)^FracPart[m]*(x^(-1))^FracPart[m]*Subst[Int[(a+b*x^(-n)+c*x^(-2*n))^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && ILtQ[n,0] && Not[RationalQ[m]]
```

7. $\int (d\,x)^m \left(a + b\,x^n + c\,x^{2\,n}\right)^p \,dx$ when $b^2 - 4\,a\,c \neq 0 \land n \in \mathbb{F}$

1: $\int x^m \left(a + b\,x^n + c\,x^{2\,n}\right)^p \,dx$ when $b^2 - 4\,a\,c \neq 0 \land n \in \mathbb{F}$

**Derivation: Integration by substitution**

**Basis: If** $k \in \mathbb{Z}^+$, **then** $x^m\,F[x^n] == k\,\text{Subst}\left[x^{k\,(m+1)-1}\,F\left[x^{k\,n}\right],\ x,\ x^{1/k}\right]\,\partial_x x^{1/k}$

**Rule 1.2.3.2.7.1: If** $b^2 - 4\,a\,c \neq 0 \land n \in \mathbb{F}$, **let** $k = \text{Denominator}[n]$, **then**

$$\int x^m \left(a + b\,x^n + c\,x^{2\,n}\right)^p dx \longrightarrow k\,\text{Subst}\left[\int x^{k\,(m+1)-1} \left(a + b\,x^{k\,n} + c\,x^{2\,k\,n}\right)^p dx,\ x,\ x^{1/k}\right]$$

**Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*x^(k*n)+c*x^(2*k*n))^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,c,m,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && FractionQ[n]
```

2: $\int (d\,x)^m \left(a + b\,x^n + c\,x^{2\,n}\right)^p \,dx$ when $b^2 - 4\,a\,c \neq 0 \land n \in \mathbb{F}$

**Derivation: Piecewise constant extraction**

**Basis:** $\partial_x \frac{(d\,x)^m}{x^m} == 0$

**Basis:** $\frac{(d\,x)^m}{x^m} == \frac{d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

**Rule 1.2.3.2.7.2: If** $b^2 - 4\,a\,c \neq 0 \land n \in \mathbb{F}$, **then**

$$\int (d\,x)^m \left(a + b\,x^n + c\,x^{2\,n}\right)^p dx \longrightarrow \frac{d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a + b\,x^n + c\,x^{2\,n}\right)^p dx$$

**Program code:**

```
Int[(d_*x_)^m_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  d^IntPart[m]*(d*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && FractionQ[n]
```

8. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,\frac{n}{m+1}\in\mathbb{Z}$

1: $\int x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,\frac{n}{m+1}\in\mathbb{Z}$

**Derivation: Integration by substitution**

- **Basis: If** $\frac{n}{m+1}\in\mathbb{Z}$**, then** $x^m\,F[x^n]=\frac{1}{m+1}\,\text{Subst}\left[F\left[x^{\frac{n}{m+1}}\right],\,x,\,x^{m+1}\right]\partial_x x^{m+1}$
- **Rule 1.2.3.2.8.1: If** $b^2-4\,a\,c\neq 0\,\wedge\,\frac{n}{m+1}\in\mathbb{Z}$

$$\int x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx\,\longrightarrow\,\frac{1}{m+1}\,\text{Subst}\left[\int\left(a+b\,x^{\frac{n}{m+1}}+c\,x^{\frac{2\,n}{m+1}}\right)^p dx,\,x,\,x^{m+1}\right]$$

**Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  1/(m+1)*Subst[Int[(a+b*x^Simplify[n/(m+1)]+c*x^Simplify[2*n/(m+1)])^p,x],x,x^(m+1)] /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

2: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,\frac{n}{m+1}\in\mathbb{Z}$

**Derivation: Piecewise constant extraction**

- **Basis:** $\partial_x \frac{(d\,x)^m}{x^m}=0$
- **Basis:** $\frac{(d\,x)^m}{x^m}=\frac{d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$
- **Rule 1.2.3.2.8.2: If** $b^2-4\,a\,c\neq 0\,\wedge\,\frac{n}{m+1}\in\mathbb{Z}$**, then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx\,\longrightarrow\,\frac{d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$$

**Program code:**

```
Int[(d_*x_)^m_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  d^IntPart[m]*(d*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

9. $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p\in\mathbb{Z}^-$

1: $\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0$

- Reference: G&R 2.161.1a

- Derivation: Algebraic expansion

- Basis: Let $q=\sqrt{b^2-4\,a\,c}$, then $\frac{1}{a+b\,z+c\,z^2}==\frac{2\,c}{q}\,\frac{1}{b-q+2\,c\,z}-\frac{2\,c}{q}\,\frac{1}{b+q+2\,c\,z}$

- Rule 1.2.3.2.9.1: If $b^2-4\,a\,c\neq 0$, let $q=\sqrt{b^2-4\,a\,c}$, then

$$\int \frac{(d\,x)^m}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,\frac{2\,c}{q}\int \frac{(d\,x)^m}{b-q+2\,c\,x^n}\,dx-\frac{2\,c}{q}\int \frac{(d\,x)^m}{b+q+2\,c\,x^n}\,dx$$

Program code:

```
Int[(d_.*x_)^m_./(a_+b_.*x_^n_+c_.*x_^n2_.),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[(d*x)^m/(b-q+2*c*x^n),x] -
  2*c/q*Int[(d*x)^m/(b+q+2*c*x^n),x]] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

2: $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,p+1\in\mathbb{Z}^-$

**Derivation: Trinomial recurrence 2b with $A=1$ and $B=0$**

**Rule 1.2.3.2.9.2: If $b^2-4\,a\,c\neq 0\,\wedge\,p+1\in\mathbb{Z}^-$, then**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$

$$-\frac{(d\,x)^{m+1}\left(b^2-2\,a\,c+b\,c\,x^n\right)\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{a\,d\,n\,(p+1)\left(b^2-4\,a\,c\right)}+$$

$$\frac{1}{a\,n\,(p+1)\left(b^2-4\,a\,c\right)}\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\left(b^2\,(m+n\,(p+1)+1)-2\,a\,c\,(m+2\,n\,(p+1)+1)+b\,c\,(m+n\,(2\,p+3)+1)\,x^n\right)dx$$

**Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  -(d*x)^(m+1)*(b^2-2*a*c+b*c*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*d*n*(p+1)*(b^2-4*a*c)) +
  1/(a*n*(p+1)*(b^2-4*a*c))*
    Int[(d*x)^m*(a+b*x^n+c*x^(2*n))^(p+1)*Simp[b^2*(n*(p+1)+m+1)-2*a*c*(m+2*n*(p+1)+1)+b*c*(2*n*p+3*n+m+1)*x^n,x],x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && ILtQ[p+1,0]
```

**10:** $\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$

**Derivation: Piecewise constant extraction**

- **Basis:** $\partial_x \frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(1+\frac{2\,c\,x^n}{b+\sqrt{b^2-4\,a\,c}}\right)^p \left(1+\frac{2\,c\,x^n}{b-\sqrt{b^2-4\,a\,c}}\right)^p} = 0$

**Rule 1.2.3.2.10:**

$$\int (d\,x)^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{a^{\text{IntPart}[p]} \left(a+b\,x^n+c\,x^{2\,n}\right)^{\text{FracPart}[p]}}{\left(1+\frac{2\,c\,x^n}{b+\sqrt{b^2-4\,a\,c}}\right)^{\text{FracPart}[p]} \left(1+\frac{2\,c\,x^n}{b-\sqrt{b^2-4\,a\,c}}\right)^{\text{FracPart}[p]}} \int (d\,x)^m \left(1+\frac{2\,c\,x^n}{b+\sqrt{b^2-4\,a\,c}}\right)^p \left(1+\frac{2\,c\,x^n}{b-\sqrt{b^2-4\,a\,c}}\right)^p dx$$

**Program code:**

```
Int[(d_.*x_)^m_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^n+c*x^(2*n))^FracPart[p]/
    ((1+2*c*x^n/(b+Rt[b^2-4*a*c,2]))^FracPart[p]*(1+2*c*x^n/(b-Rt[b^2-4*a*c,2]))^FracPart[p])*
    Int[(d*x)^m*(1+2*c*x^n/(b+Sqrt[b^2-4*a*c]))^p*(1+2*c*x^n/(b-Sqrt[b^2-4*a*c]))^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[n2,2*n]
```

**11.** $\int (d\,x)^m \left(a + b\,x^{-n} + c\,x^n\right)^p\,dx$

**1.** $\int x^m \left(a + b\,x^{-n} + c\,x^n\right)^p\,dx$

**1:** $\int x^m \left(a + b\,x^{-n} + c\,x^n\right)^p\,dx$ when $p \in \mathbb{Z}$

**Derivation: Algebraic normalization**

**Basis:** $a + b\,x^{-n} + c\,x^n == x^{-n} \left(b + a\,x^n + c\,x^{2\,n}\right)$

**Rule 1.2.3.2.11.1.1: If** $p \in \mathbb{Z}$**, then**

$$\int x^m \left(a + b\,x^{-n} + c\,x^n\right)^p\,dx \longrightarrow \int x^{m-n\,p} \left(b + a\,x^n + c\,x^{2\,n}\right)^p\,dx$$

**Program code:**

```
Int[x_^m_.*(a_+b_.*x_^mn_+c_.*x_^n_.)^p_.,x_Symbol] :=
  Int[x^(m-n*p)*(b+a*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,m,n},x] && EqQ[mn,-n] && IntegerQ[p] && PosQ[n]
```

**2:** $\int x^m \left(a + b\,x^{-n} + c\,x^n\right)^p\,dx$ when $p \notin \mathbb{Z}$

**Derivation: Piecewise constant extraction**

- **Basis:** $\partial_x \frac{x^{n\,p} \left(a+b\,x^{-n}+c\,x^n\right)^p}{\left(b+a\,x^n+c\,x^{2\,n}\right)^p} == 0$
- **Basis:** $\frac{x^{n\,p} \left(a+b\,x^{-n}+c\,x^n\right)^p}{\left(b+a\,x^n+c\,x^{2\,n}\right)^p} == \frac{x^{n\,\text{FracPart}[p]} \left(a+b\,x^{-n}+c\,x^n\right)^{\text{FracPart}[p]}}{\left(b+a\,x^n+c\,x^{2\,n}\right)^{\text{FracPart}[p]}}$

**Rule 1.2.3.2.11.1.2: If** $p \notin \mathbb{Z}$**, then**

$$\int x^m \left(a + b\,x^{-n} + c\,x^n\right)^p\,dx \longrightarrow \frac{x^{n\,\text{FracPart}[p]} \left(a + b\,x^{-n} + c\,x^n\right)^{\text{FracPart}[p]}}{\left(b + a\,x^n + c\,x^{2\,n}\right)^{\text{FracPart}[p]}} \int x^{m-n\,p} \left(b + a\,x^n + c\,x^{2\,n}\right)^p\,dx$$

**Program code:**

```
Int[x_^m_.*(a_+b_.*x_^mn_+c_.*x_^n_.)^p_.,x_Symbol] :=
  x^(n*FracPart[p])*(a+b/x^n+c*x^n)^FracPart[p]/(b+a*x^n+c*x^(2*n))^FracPart[p]*Int[x^(m-n*p)*(b+a*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[mn,-n] && Not[IntegerQ[p]] && PosQ[n]
```

**2:** $\int (d\,x)^m \left(a + b\,x^{-n} + c\,x^n\right)^p \,dx$

**Derivation: Piecewise constant extraction**

- **Basis:** $\partial_x \frac{(d\,x)^m}{x^m} == 0$
- **Basis:** $\frac{(d\,x)^m}{x^m} == \frac{d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

**Rule 1.2.3.2.11.2:**

$$\int (d\,x)^m \left(a + b\,x^{-n} + c\,x^n\right)^p \,dx \longrightarrow \frac{d^{\text{IntPart}[m]}\,(d\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a + b\,x^{-n} + c\,x^n\right)^p \,dx$$

- **Program code:**

```
Int[(d_*x_)^m_.*(a_+b_.*x_^mn_+c_.*x_^n_.)^p_.,x_Symbol] :=
  d^IntPart[m]*(d*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^(-n)+c*x^n)^p,x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[mn,-n]
```

**S.** $\int u^m \left(a + b\,v^n + c\,v^{2\,n}\right)^p \,dx$ when $v == d + e\,x \wedge u == f\,v$

**1:** $\int x^m \left(a + b\,v^n + c\,v^{2\,n}\right)^p \,dx$ when $v == d + e\,x \wedge m \in \mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis:** If $m \in \mathbb{Z}$, then $x^m\,F[d + e\,x] == \frac{1}{e^{m+1}}\,\text{Subst}[(x - d)^m\,F[x],\,x,\,d + e\,x]\,\partial_x (d + e\,x)$
- **Rule 1.2.3.2.S.1:** If $v == d + e\,x \wedge m \in \mathbb{Z}$, then

$$\int x^m \left(a + b\,v^n + c\,v^{2\,n}\right)^p \,dx \longrightarrow \frac{1}{e^{m+1}}\,\text{Subst}\left[\int (x - d)^m \left(a + b\,x^n + c\,x^{2\,n}\right)^p \,dx,\,x,\,v\right]$$

- **Program code:**

```
Int[x_^m_.*(a_.+b_.*v_^n_+c_.*v_^n2_.)^p_.,x_Symbol] :=
  1/Coefficient[v,x,1]^(m+1)*Subst[Int[SimplifyIntegrand[(x-Coefficient[v,x,0])^m*(a+b*x^n+c*x^(2*n))^p,x],x],x,v] /;
FreeQ[{a,b,c,n,p},x] && EqQ[n2,2*n] && LinearQ[v,x] && IntegerQ[m] && NeQ[v,x]
```

2: $\int u^m \left(a + b\, v^n + c\, v^{2n}\right)^p \, dx$ when $v == d + e\, x \wedge u == f\, v$

- **Derivation: Integration by substitution and piecewise constant extraction**
- **Basis: If $u == f\, v$, then $\partial_x \frac{u^m}{v^m} == 0$**
- **Rule 1.2.3.2.S.2: If $v == d + e\, x \wedge u == f\, v$, then**

$$\int u^m \left(a + b\, v^n + c\, v^{2n}\right)^p \, dx \longrightarrow \frac{u^m}{e\, v^m} \, \text{Subst}\left[\int x^m \left(a + b\, x^n + c\, x^{2n}\right)^p \, dx,\ x,\ v\right]$$

- **Program code:**

```
Int[u_^m_.*(a_.+b_.*v_^n_+c_.*v_^n2_.)^p_.,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*(a+b*x^n+c*x^(2*n))^p,x],x,v] /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[n2,2*n] && LinearPairQ[u,v,x]
```